Math 277
Ordinary Differential Equations

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 8

Second Order Homogeneous Linear Equations with Constant Coefficients

Initializations

```
> restart;
```

8.1 Second Order Homogeneous Linear Equations with Constant Coefficients.

The form of the general solution of the equation

$$a\,y''+b\,y'+c\,y=0$$

depends on the number and the type of the roots of the corresponding auxiliary equation

$$ar^2+br+c=0$$

We distinguish three cases.

1. The auxiliary equation has two distinct real roots.
2. The auxiliary equation has a repeated real root.
3. The auxiliary equation has two conjugate complex roots.

The examples below highlight the various possibilities. Mathematical details will be provided in class.

Examples

Example 8.1.1

Solve the initial value problem

$$y''+2\,y'-y=0; \quad y(0)=0,\ y'(0)=-1$$

Solution
Code the differential equation and generate the auxiliary equation.

```
> deq:=diff(y(t), t$2)+2*diff(y(t), t)-y(t)=0;
```

$$deq := \frac{d^2}{dt^2}\,y(t)+2\left(\frac{d}{dt}\,y(t)\right)-y(t)=0 \qquad \textbf{(2.1.1.1)}$$

```
> e1:=eval(deq, y(t)=exp(r*t));
  aux:=expand(e1/exp(r*t));
```

$$e1 := r^2 e^{rt} + 2\, r\, e^{rt} - e^{rt} = 0$$

$$aux := r^2 + 2\, r - 1 = 0 \tag{2.1.1.2}$$

Compute the eigenvalues.

```
> ev:=solve(aux, r);
```

$$ev := -1 + \sqrt{2},\, -1 - \sqrt{2} \tag{2.1.1.3}$$

Note that the eigenvalues could have been computed directly by solving the equation

$$r^2 e^{rt} + 2\, r\, e^{rt} - e^{rt} = 0$$

for the parameter *r*.

```
> ev:=solve(e1, r);
```

$$ev := -1 + \sqrt{2},\, -1 - \sqrt{2} \tag{2.1.1.4}$$

Construct the general solution and implement the initial conditions.

```
> gs:=add(c[k]*exp(ev[k]*t), k=1..2);
```

$$gs := c_1\, e^{\left(-1 + \sqrt{2}\right) t} + c_2\, e^{\left(-1 - \sqrt{2}\right) t} \tag{2.1.1.5}$$

```
> eq1:=eval(gs, t=0)=0;
  eq2:=eval(diff(gs, t), t=0)=-1;
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 + c_2 = 0$$

$$eq2 := c_1 \left(-1 + \sqrt{2}\right) + c_2 \left(-1 - \sqrt{2}\right) = -1$$

$$val\_c := \left\{c_1 = -\frac{1}{4}\sqrt{2},\, c_2 = \frac{1}{4}\sqrt{2}\right\} \tag{2.1.1.6}$$

Hence, the solution to the given initial value problem equals

```
> sol:=eval(gs, val_c);
```

$$sol := -\frac{1}{4}\sqrt{2}\, e^{\left(-1 + \sqrt{2}\right) t} + \frac{1}{4}\sqrt{2}\, e^{\left(-1 - \sqrt{2}\right) t} \tag{2.1.1.7}$$

Example 8.1.2

Solve the initial value problem

$$y'' + 8\, y' + 16\, y = 0; \quad y(0) = -3,\ y'(0) = 2$$

Solution

Code the differential equation and generate the auxiliary equation..

```
> deq:=diff(y(t), t$2)+8*diff(y(t), t)+16*y(t)=0;
```

$$deq := \frac{d^2}{dt^2} y(t) + 8 \left(\frac{d}{dt} y(t)\right) + 16\, y(t) = 0 \tag{2.1.2.1}$$

```
> e1:=eval(deq, y(t)=exp(r*t));
  aux:=expand(e1/exp(r*t));
```

$$e1 := r^2 e^{rt} + 8\, r\, e^{rt} + 16\, e^{rt} = 0$$

$$aux := r^2 + 8r + 16 = 0 \quad \textbf{(2.1.2.2)}$$

Compute the eigenvalues.

```
> ev:=solve(aux, r);
```

$$ev := -4, -4 \quad \textbf{(2.1.2.3)}$$

Since we have a repeated eigenvalue the general solution is given by

```
> gs:=c[1]*exp(ev[1]*t)+c[2]*t*exp(ev[1]*t);
```

$$gs := c_1 e^{-4t} + c_2 t e^{-4t} \quad \textbf{(2.1.2.4)}$$

Next we implement the initial conditions and assemble the solution of the initial value problem.

```
> eq1:=eval(gs, t=0)=-3;
  eq2:=eval(diff(gs, t), t=0)=2;
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 = -3$$

$$eq2 := -4 c_1 + c_2 = 2$$

$$val\_c := \{c_1 = -3, c_2 = -10\} \quad \textbf{(2.1.2.5)}$$

Hence, the solution to the given initial value problem equals

```
> sol:=eval(gs, val_c);
```

$$sol := -3 e^{-4t} - 10 t e^{-4t} \quad \textbf{(2.1.2.6)}$$

Example 8.1.3

Solve the initial value problem

$$y'' + 2y' + 5y = 0; \quad y(0) = -1, \ y'(0) = 4$$

Solution

Code the differential equation and generate the auxiliary equation.

```
> deq:=diff(y(t), t$2)+2*diff(y(t), t)+5*y(t)=0;
```

$$deq := \frac{d^2}{dt^2} y(t) + 2\left(\frac{d}{dt} y(t)\right) + 5 y(t) = 0 \quad \textbf{(2.1.3.1)}$$

```
> e1:=eval(deq, y(t)=exp(r*t));
  aux:=expand(e1/exp(r*t));
```

$$e1 := r^2 e^{rt} + 2 r e^{rt} + 5 e^{rt} = 0$$

$$aux := r^2 + 2r + 5 = 0 \quad \textbf{(2.1.3.2)}$$

Compute the eigenvalues.

```
> ev:=solve(aux, r);
```

$$ev := -1 + 2I, -1 - 2I \quad \textbf{(2.1.3.3)}$$

Note that the eigenvalues are complex, so the general solution is given by

```
> gs:=c[1]*exp(Re(ev[1])*t)*cos(Im(ev[1])*t)+c[2]*exp(Re(ev
  [1])*t)*sin(Im(ev[1])*t);
```

$$gs := c_1 e^{-t} \cos(2t) + c_2 e^{-t} \sin(2t) \quad \textbf{(2.1.3.4)}$$

Implement the initial conditions.

```
> eq1:=eval(gs, t=0)=-1;
  eq2:=eval(diff(gs, t), t=0)=4;
  val_c:=solve({eq1, eq2}, {c[1], c[2]});
```

$$eq1 := c_1 = -1$$

$$eq2 := -c_1 + 2\,c_2 = 4$$

$$val\_c := \left\{c_1 = -1, c_2 = \frac{3}{2}\right\} \quad \textbf{(2.1.3.5)}$$

Hence, the solution to the given initial value problem equals

```
> sol:=eval(gs, val_c);
```

$$sol := -e^{-t} \cos(2t) + \frac{3}{2} e^{-t} \sin(2t) \quad \textbf{(2.1.3.6)}$$